MetaL — A Library for Formalised Metatheory in Agda

Robin Adams[1]

University of Bergen `r.adams@ii.uib.no`

Abstract. There are now many techniques for formalising metatheory (nominal sets, higher-order abstract syntax, etc.) but, in general, each requires the syntax and rules of deduction for a system to be defined afresh, and so all the proofs of basic lemmas must be written anew when we work with a new system, and modified every time we modify the system.

In this rough diamond, we present an early version of MetaL ("Metatheory Library"), a library for formalised metatheory in Agda. There is a type Grammar of grammars with binding, and types Red G of reduction relations and Rule G of sets of rules of deduction over G : Grammar. A grammar is given by a set of constructors, whose type specifies how many arguments it takes, and how many variables are bound in each argument. Reduction relations and rules of deduction are given by patterns, or expressions involving second-order variables.

The library includes a general proof of the substitution lemma. The final version is planned to include proofs of Church-Rosser for reductions with no critical pairs, and Weakening and Substitution lemmas for appropriate sets of rules of deduction.

MetaL has been designed with the following criteria in mind. It is easy to specify a grammar, reduction rule or set of rules of deduction: the Agda definition is the same length as the definition on paper. The general results are immediately applicable. When working within a grammar G, it should be possible to define functions by induction on expressions, and prove results by induction on expressions or induction on derivations, using only Agda's built-in pattern matching.

1 Introduction

1.1 Design Criteria

This library was produced with the following design goals.

– The library should be *modular*. There should be a type Grammar, and results such as the Substitution Lemma should be provable 'once and for all' for all grammars.[1]

[1] For future versions of the library, we wish to have a type of reduction rules over a grammar, and a type of theories (sets of rules of deduction) over a grammar.

- It should be possible for the user to define their own operations, such as path substitution
- Operations which are defined by induction on expressions should be definable by induction in Agda. Results which are proved by induction on expressions should be proved by induction in Agda.

2 Grammar

Example 1 (Simply Typed Lambda Calculus). For a running example, we will construct the grammar of the simply-typed lambda-calculus, with Church-typing and one constant ground type $\perp$. On paper, in BNF-style, we write the grammar as follows:

$$\begin{array}{ll} \text{Type} & A ::= \perp \mid A \to A \\ \text{Term} & M ::= x \mid MM \mid \lambda x : A.M \end{array}$$

2.1 Taxonomy

A *taxonomy* is a set of *expression kinds*, divided into *variable kinds* and *non-variable kinds*. The intention is that the expressions of the grammar are divided into expression kinds. Every variable ranges over the expressions of one (and only one) variable kind.

```
record Taxonomy : Set₁ where
 field
  VariableKind : Set
  NonVariableKind : Set

 data ExpressionKind : Set where
  varKind : VariableKind → ExpressionKind
  nonVariableKind : NonVariableKind → ExpressionKind
```

An *alphabet* is a finite set of *variables*, to each of which is associated a variable kind. We write Var V K for the set of all variables in the alphabet V of kind K.

```
 infixl 55 _,_
 data Alphabet : Set where
  ∅ : Alphabet
  _,_ : Alphabet → VariableKind → Alphabet


 data Var : Alphabet → VariableKind → Set where
  x₀ : ∀ {V} {K} → Var (V , K) K
  ↑ : ∀ {V} {K} {L} → Var V L → Var (V , K) L
```

Example 2. For the simply-typed lambda-calculus, there are two expression kinds: type, which is a non-variable kind, and term, which is a variable kind:

```
data stlcVariableKind : Set where
 -term : stlcVariableKind

data stlcNonVariableKind : Set where
 -type : stlcNonVariableKind

stlcTaxonomy : Taxonomy
stlcTaxonomy = record {
 VariableKind = stlcVariableKind ;
 NonVariableKind = stlcNonVariableKind }
```

2.2 Grammar

Definition 3. *An* abstraction kind *has the form* $K_1 \to \cdots \to K_n \to L$*, where each* K_i *is an abstraction kind, and* L *is an expression kind.*

A constructor kind *has the form* $A_1 \to \cdots \to A_n \to K$*, where each* A_i *is an abstraction kind, and* K *is an expression kind.*

To define these, we introduce the notion of a *simple kind*: a simple kind over sets S and T is an object of the form $s_1 \to \cdots \to s_n \to t$, where each $s_i \in S$ and $t \in T$.

We implement this by saying a simple kind over S and T consists of a list of objects of S, and one object of T:

```
record SimpleKind (A B : Set) : Set where
 constructor SK
 field
  dom : List A
  cod : B

infix 71 _◊
_◊ : ∀ {A} {B} → B → SimpleKind A B
b ◊ = SK [] b

infixr 70 _⟶_
_⟶_ : ∀ {A} {B} → A → SimpleKind A B → SimpleKind A B
a ⟶ SK dom cod = SK (a :: dom) cod
```

We can construct an object of type SK S T by writing

$$s_1 \longrightarrow \cdots \longrightarrow s_n \longrightarrow t\Diamond \enspace .$$

(The '◊' symbol marks the end of the simple kind. It is needed to help Agda disambiguate the syntax.)

We are now able to write Definition 3 like this:

```
AbstractionKind = SimpleKind VariableKind ExpressionKind
ConstructorKind = SimpleKind AbstractionKind ExpressionKind
```

A *grammar* over a taxonomy consists of:

- a set of *constructors*, each with an associated constructor kind;
- a function assigning, to each variable kind, an expression kind, called its *parent*. (The intention is that, when a declaration $x : A$ occurs in a context, if x has kind K, then the kind of A is the parent of K.)

```
record IsGrammar (T : Taxonomy) : Set₁ where
 open Taxonomy T
 field
  Con : ConstructorKind → Set
  parent : VariableKind → ExpressionKind

record Grammar : Set₁ where
 field
  taxonomy : Taxonomy
  isGrammar : IsGrammar taxonomy
 open Taxonomy taxonomy public
 open IsGrammar isGrammar public
```

Definition 4. *We define simultaneously the set of* expressions *of kind K over V for every expression kind K and alphabet V; and the set of* abstractions *of kind A over V for every abstraction kind A and alphabet V.*

- *Every variable of kind K in V is an expression of kind K over V.*
- *If c is a constructor of kind $A_1 \to \cdots \to A_n \to K$, and M_1 is an abstraction of kind A_1, ..., M_n is an abstraction of kind A_n (all over V), then*

$$cM_1 \cdots M_n$$

  is an expression of kind K over V.
- *An abstraction of kind $K_1 \to \cdots \to K_n \to L$ over V is an expression of the form*

$$[x_1, \ldots, x_n]M$$

  where each x_i is a variable of kind K_i, and M is an expression of kind L over $V \cup \{x_1, \ldots, x_n\}$.

In the Agda code, we define simultaneously the following four types:

- Expression $V\,K$ = Subexp V-Expression K, the type of expressions of kind K;
- VExpression $V\,K$ = Expression V (varKind K), a convenient shorthand when K is a variable kind;

– Abstraction $V\,A$, the type of abstractions of kind A over V
– ListAbstraction $V\,AA$: if $AA \equiv [A_1, \ldots, A_n]$, then ListAbstraction $V\,AA$ is the type of lists of abstractions $[M_1, \ldots, M_n]$ such that each M_i is of kind A_i.

```
data Subexp (V : Alphabet) : ∀ C → Kind C → Set
Expression : Alphabet → ExpressionKind → Set
VExpression : Alphabet → VariableKind → Set
Abstraction : Alphabet → AbstractionKind → Set
ListAbstraction : Alphabet → List AbstractionKind → Set

Expression V K = Subexp V -Expression K
VExpression V K = Expression V (varKind K)
Abstraction V (SK KK L) = Expression (extend V KK) L
ListAbstraction V AA = Subexp V -ListAbstraction AA

infixr 5 _::_
data Subexp V where
 var : ∀ {K} → Var V K → VExpression V K
 app : ∀ {AA} {K} → Con (SK AA K) → ListAbstraction V AA → Expression V K
 [] : ListAbstraction V []
 _::_ : ∀ {A} {AA} → Abstraction V A → ListAbstraction V AA → ListAbstraction V (A :: AA)
```

Example 5. The grammar given in Example 1 has four constructors:

– $\bot$, of kind type;
– $\to$, of kind type $\longrightarrow$ type $\longrightarrow$ type
– appl, of kind term $\longrightarrow$ term $\longrightarrow$ term
– λ, of kind type $\longrightarrow$ (term $\longrightarrow$ term) $\longrightarrow$ term

The kind of the final constructor λ should be read like this: λ takes a type A and a term M, binds a term variable x within M, and returns a term $\lambda x : A.M$

```
type : ExpressionKind
type = nonVariableKind -type

term : ExpressionKind
term = varKind -term

data stlcCon : ConstructorKind → Set where
 -bot : stlcCon (type ◊)
 -arrow : stlcCon (type ◊ ⟶ type ◊ ⟶ type ◊)
 -app : stlcCon (term ◊ ⟶ term ◊ ⟶ term ◊)
 -lam : stlcCon (type ◊ ⟶ (-term ⟶ term ◊) ⟶ term ◊)

stlcParent : VariableKind → ExpressionKind
stlcParent -term = type
```

```
stlc : Grammar
stlc = record {
 taxonomy = stlcTaxonomy ;
 isGrammar = record {
  Con = stlcCon ;
  parent = stlcParent } }

Type : Alphabet → Set
Type V = Expression V type

Term : Alphabet → Set
Term V = Expression V term

⊥ : ∀ V → Type V
⊥ V = app -bot []

_⇒_ : ∀ {V} → Type V → Type V → Type V
A ⇒ B = app -arrow (A :: B :: [])

appl : ∀ {V} → Term V → Term V → Term V
appl M N = app -app (M :: N :: [])

Λ : ∀ {V} → Type V → Term (V , -term) → Term V
Λ A M = app -lam (A :: M :: [])
```

2.3 Families of Operations

Our next aim is to define replacement and substitution. Many of the results about these two operations have very similar proofs, so in order to avoid duplicating code, we make the following definition.

Definition 6 (Family of Operations). *A* family of operations $\Rightarrow$ *consists of:*

- *for any alphabets U, V, a set $U \Rightarrow V$ of* operations *from U to V;*
- *for any operation $\sigma : U \Rightarrow V$ and variable x : Var U K, an expression $\sigma(x)$: Expression V K*
- *for any alphabet V and variable kind K, an operation $\uparrow : V \Rightarrow V, K$*
- *for any alphabet V, an operation $1_V : V \Rightarrow V$*
- *for any operations $\rho : U \Rightarrow V$ and $\sigma : V \Rightarrow W$, an operation $\sigma \circ \rho : U \Rightarrow W$, the* composition *of σ and ρ;*
- *for any operation $\sigma : U \Rightarrow V$ and variable kind K, an operation $\sigma^{\uparrow} : U, K \Rightarrow V, K$, the* lifting *of σ;*

such that:

- $\uparrow(x) \equiv x$ *for any variable* x
- $1_V(x) \equiv x$ *for any variable* x
- $\sigma^{\uparrow}(x_0) \equiv x_0$
- $\sigma^{\uparrow}(x) \equiv \sigma(x)[\uparrow]$
- $(\sigma \circ \rho)(x) \equiv \rho(x)[\sigma]$

where, if E : $\mathsf{Expression}\, U\, K$ *and* $\sigma : U \Rightarrow V$ *then* $E[\sigma]$: $\mathsf{Expression}\, V\, K$*, the* action *of* σ *on* E*, is defined by*

$$\begin{aligned} x[\sigma] &\stackrel{\text{def}}{=} \sigma(x) \\ ([x_1, \ldots, x_n]E)[\sigma] &\stackrel{\text{def}}{=} E[\sigma^{\uparrow\uparrow\cdots\uparrow}] \\ (cE_1 \cdots E_n)[\sigma] &\stackrel{\text{def}}{=} c(E_1[\sigma]) \cdots (E_n[\sigma]) \end{aligned}$$

We write $\rho \sim \sigma$ *iff* ρ *and* σ *are extensionally equal, i.e.* $\rho(x) \equiv \sigma(x)$ *for every variable* x*.*

The way that this is formalised in Agda is described in Appendix A.

It is easy to see that our two examples of replacement and substitution fit this pattern.

Definition 7 (Replacement). Replacement *is the family of operations defined as follows.*

- *A* replacement *from* U *to* V*,* $\rho : U \rightarrow_R V$*, is a family of functions* ρ_K : $\mathsf{Var}\, U\, K \rightarrow \mathsf{Var}\, V\, K$ *for every variable kind* K*.*
- *For* x : $\mathsf{Var}\, U\, K$*, define* $\rho(x) \stackrel{\text{def}}{=} \rho_K(x)$*.*
- *Define* $\uparrow : V \rightarrow_R V, K$ *by* $\uparrow_L(x) \equiv x$*.*
- *Define* $(1_V)_K(x) \equiv x$
- *Define* $(\sigma \circ \rho)_K(x) \equiv \sigma_K(\rho_K(x))$
- *Define* $\sigma^{\uparrow}_K(x_0) \equiv x_0$*, and* $\sigma^{\uparrow}_L(\uparrow x) \equiv \uparrow \sigma_L(x)$*.*

```
REP : OpFamily
```

We write $E\langle\rho\rangle$ for the action of a replacement ρ on a subexpression E.

Definition 8 (Substitution). Substitution *is the family of operations defined as follows.*

- *A* substitution *from* U *to* V*,* $\sigma : U \Rightarrow V$*, is a family of functions* σ_K : $\mathsf{Var}\, U\, K \rightarrow \mathsf{Expression}\, V\, K$ *for every variable kind* K*.*
- *For* x : $\mathsf{Var}\, U\, K$*, define* $\sigma(x) \stackrel{\text{def}}{=} \sigma_K(x)$
- *Define* $\uparrow : V \rightarrow_R V, K$ *by* $\uparrow_L(x) \equiv x$*.*
- *Define* $(1_V)_K(x) \equiv x$
- *Define* $(\sigma \circ \rho)_K(x) \equiv \rho_K(x)[\sigma]$
- *Define* $\sigma^{\uparrow}_K(x_0) \equiv x_0$ *and* $\sigma^{\uparrow}_L(\uparrow x) \equiv \sigma_L(x)\langle\uparrow\rangle$*.*

```
SUB : OpFamily
```

We write $E[\sigma]$ for the action of a substitution σ on a subexpression E.

Results about Families of Operations. We can prove the following results about an arbitrary family of operations.

Lemma 9. *1. If $\rho \sim \sigma$ then $E[\rho] \equiv E[\sigma]$.*

```
ap-congl : ∀ {U} {V} {C} {K}
  {ρ σ : Op U V} → ρ ∼op σ → ∀ (E : Subexp U C K) →
  ap ρ E ≡ ap σ E
```

2. $1_V^{\uparrow} = 1_{V,K}$

```
liftOp-idOp : ∀ {V} {K} → liftOp K (idOp V) ∼op idOp (V , K)
```

3. $E[1_V] \equiv E$

```
ap-idOp : ∀ {V} {C} {K} {E : Subexp V C K} → ap (idOp V) E ≡ E
```

4. $E[\sigma \circ \rho] \equiv E[\rho][\sigma]$

```
ap-comp : ∀ {U V W C K} (E : Subexp U C K) {σ ρ} →
  ap H (_∘_ {U} {V} {W} σ ρ) E ≡ ap F σ (ap G ρ E)
```

5. $\tau \circ (\sigma \circ \rho) \sim (\tau \circ \sigma) \circ \rho$

```
assoc : ∀ {U} {V} {W} {X}
  {τ : Op W X} {σ : Op V W} {ρ : Op U V} →
  τ ∘ (σ ∘ ρ) ∼op (τ ∘ σ) ∘ ρ
```

6. If $\sigma : U \Rightarrow V$ then $1_V \circ \sigma \sim \sigma \sim \sigma \circ 1_U$

```
unitl : ∀ {U} {V} {σ : Op U V} → idOp V ∘ σ ∼op σ

unitr : ∀ {U} {V} {σ : Op U V} → σ ∘ idOp U ∼op σ
```

2.4 Substitution for the Last Variables

Given an alphabet $V \cup \{x_0, \ldots, x_n\}$ and expressions $E_0, \ldots, E_n$, we define the substitution

$$[x_0 := E_0, \ldots, x_n := E_n] : V \cup \{x_0, \ldots, x_n\} \Rightarrow V$$

```
botSub : ∀ {V} {KK} → HetsnocList (VExpression V) KK → Sub (snoc-extend V KK) V
botSub {KK = []} _ x = var x
botSub {KK = _ snoc _} (_ snoc E) x₀ = E
botSub {KK = _ snoc _} (EE snoc _) (↑ x) = botSub EE x

infix 65 x₀:=_
x₀:=_ : ∀ {V} {K} → Expression V (varKind K) → Sub (V , K) V
x₀:= E = botSub ([] snoc E)
```

We have the following results about this substitution:

Lemma 10. *1.* $E'\langle\uparrow\rangle[x_0 := E] \equiv E$

> botSub-up : ∀ {F} {V} {K} {C} {L} {E : Expression V (varKind K)} (comp : Composition SubLF F S
> ap F (up F) E' ⟦ x₀:= E ⟧ ≡ E'

2. $E'[x_0 := E][\sigma] \equiv E'[\sigma^{\uparrow}][x_0 := E[\sigma]]$

> •-botSub'' : ∀ {U} {V} {C} {K} {L}
> {E : Expression U (varKind K)} {σ : Sub U V} (F : Subexp (U , K) C L) →
> F ⟦ x₀:= E ⟧ ⟦ σ ⟧ ≡ F ⟦ liftSub K σ ⟧ ⟦ x₀:= (E ⟦ σ ⟧) ⟧

3 Limitations

– There is no way to express that an expression depends on some variable kinds but not others. (E.g. in our simply-typed lambda calculus example: the types do not depend on the term variables.) This leads to some boilerplate that is needed, proving lemmas of the form

$$(\perp U)[\sigma] \equiv \perp V \tag{1}$$

There is a workaround for this special case. We can declare all the types as constants: This is what we used for the project PHOML.
For a general solution, we would need to parametrise alphabets by the set of variable kinds that may occur in them, and then prove results about mappings from one type of alphabet to another. We could then prove once-and-for-all versions of the lemmas like (1). It remains to be seen whether this would still be unwieldy in practice.

A Formalisation of Families of Operations

We define the type of families of operations in several stages, as follows.

A.1 Pre-family of Operations

Definition 11 (Pre-family of Operations). *A* pre-family of operations $\Rightarrow$ *consists of:*

–
– *for any alphabets U, V, a set $U \Rightarrow V$ of* operations *from U to V;*
– *for any operation $\sigma : U \Rightarrow V$ and variable x : Var$U\,K$, an expression $\sigma(x)$: Expression$\,V\,K$*
– *for any alphabet V and variable kind K, an operation $\uparrow : V \Rightarrow V, K$*
– *for any alphabet V, an operation $1_V : V \Rightarrow V$*

such that:

- $\uparrow(x) \equiv x$
- $1_V(x) \equiv x$

```
record PreOpFamily : Set₂ where
 field
  Op : Alphabet → Alphabet → Set
  apV : ∀ {U} {V} {K} → Op U V → Var U K → Expression V (varKind K)
  up : ∀ {V} {K} → Op V (V , K)
  apV-up : ∀ {V} {K} {L} {x : Var V K} → apV (up {K = L}) x ≡ var (↑ x)
  idOp : ∀ V → Op V V
  apV-idOp : ∀ {V} {K} (x : Var V K) → apV (idOp V) x ≡ var x
```

Let $\Rightarrow$ be a pre-family of operations.

Definition 12. *Two operations* $\rho, \sigma : U \Rightarrow V$ *are* extensionally equal*,* $\rho \sim \sigma$*, iff* $\rho(x) \equiv \sigma(x)$ *for all variables* x*.*

We prove that this is an equivalence relation.

```
_∼op_ : ∀ {U} {V} → Op U V → Op U V → Set
_∼op_ {U} {V} ρ σ = ∀ {K} (x : Var U K) → apV ρ x ≡ apV σ x

∼-refl : ∀ {U} {V} {σ : Op U V} → σ ∼op σ

∼-sym : ∀ {U} {V} {σ τ : Op U V} → σ ∼op τ → τ ∼op σ

∼-trans : ∀ {U} {V} {ρ σ τ : Op U V} → ρ ∼op σ → σ ∼op τ → ρ ∼op τ
```

A.2 Lifting

Definition 13 (Lifting). *A* lifting *on a pre-family of operations is a mapping that, given an operation* $\rho : U \Rightarrow V$ *and variable kind* K*, returns an operation* $\rho^{\uparrow} : U, K \Rightarrow V, K$*.*

```
record Lifting (F : PreOpFamily) : Set₁ where
 open PreOpFamily F
 field
  liftOp : ∀ {U} {V} A → Op U V → Op (U , A) (V , A)
  liftOp-cong : ∀ {V} {W} {A} {ρ σ : Op V W} → ρ ∼op σ → liftOp A ρ ∼op liftOp A σ
```

Given a pre-family of operations and a lifting, we can define the action of an operation on a subexpression:

$$
\begin{aligned}
x[\sigma] &\stackrel{\text{def}}{=} \sigma(x) \\
([x_1, \ldots, x_n]E)[\sigma] &\stackrel{\text{def}}{=} E[\sigma^{\uparrow\uparrow\cdots\uparrow}] \\
(cE_1 \cdots E_n)[\sigma] &\stackrel{\text{def}}{=} c(E_1[\sigma]) \cdots (E_n[\sigma])
\end{aligned}
$$

```
ap : ∀ {U} {V} {C} {K} → Op U V → Subexp U C K → Subexp V C K
ap ρ (var x) = apV ρ x
ap ρ (app c EE) = app c (ap ρ EE)
ap _ [] = []
ap ρ (_::_ {A = SK AA _} E EE) = ap (liftsOp AA ρ) E :: ap ρ EE
```

A.3 Pre-family of Operations with Lifting

Definition 14 (Pre-family of Operations with Lifting). *A* pre-family of operations with lifting *is given by a pre-family of operations* $\Rightarrow$ *and a lifting* $^\uparrow$ *such that:*

- $\sigma^\uparrow(x_0) \equiv x_0$
- $\sigma^\uparrow(\uparrow x) \equiv \sigma(x)[\uparrow]$

```
record IsLiftFamily (F : PreOpFamily) (L : Lifting F) : Set₁ where
 open PreOpFamily F
 open Lifting L
 field
  liftOp-x₀ : ∀ {U} {V} {K} {σ : Op U V} →
   apV (liftOp K σ) x₀ ≡ var x₀
  liftOp-↑ : ∀ {U} {V} {K} {L} {σ : Op U V} (x : Var U L) →
   apV (liftOp K σ) (↑ x) ≡ ap up (apV σ x)
```

A.4 Composition

Definition 15 (Composition). *Let* $\Rightarrow_1$, $\Rightarrow_2$ *and* $\Rightarrow_3$ *be pre-families of operations with liftings. A* composition $\circ : (\Rightarrow_1); (\Rightarrow_2) \to (\Rightarrow_3)$ *is a family of mappings*

$$\circ_{UVW} : (V \Rightarrow_1 W) \times (U \Rightarrow_2 V) \to (U \Rightarrow_3 W)$$

for all alphabets U, V, W *such that:*

$$
\begin{aligned}
(\sigma \circ \rho)^\uparrow &\sim \sigma^\uparrow \circ \rho^\uparrow \\
(\sigma \circ \rho)(x) &\equiv \rho(x)[\sigma]
\end{aligned}
$$

for all ρ, σ, x.

```
record Composition (F G H : LiftFamily) : Set where
 infix 25 _∘_
 field
  _∘_ : ∀ {U} {V} {W} → Op F V W → Op G U V → Op H U W
  liftOp-comp' : ∀ {U V W K σ ρ} →
   _∼op_ H (liftOp H K (_∘_ {U} {V} {W} σ ρ))
    (liftOp F K σ ∘ liftOp G K ρ) - TODO Prove this
  apV-comp : ∀ {U} {V} {W} {K} {σ} {ρ} {x : Var U K} →
   apV H (_∘_ {U} {V} {W} σ ρ) x ≡ ap F σ (apV G ρ x)
```

Let us write $[\ ]_1$, $[\ ]_2$, $[\ ]_3$ for the action of $\Rightarrow_1$, $\Rightarrow_2$, $\Rightarrow_3$ respectively.

Lemma 16. *If $\circ$ is a composition, then $E[\sigma \circ \rho]_3 \equiv E[\rho]_2[\sigma]_1$.*

```
 ap-comp : ∀ {U V W C K} (E : Subexp U C K) {σ ρ} →
  ap H (_∘_ {U} {V} {W} σ ρ) E ≡ ap F σ (ap G ρ E)
```

Lemma 17. *Let $\Rightarrow_1$, $\Rightarrow_2$, $\Rightarrow_3$, $\Rightarrow_4$ be pre-families of operations with liftings. Suppose there exist compositions $(\Rightarrow_1); (\Rightarrow_2) \to (\Rightarrow_4)$ and $(\Rightarrow_2); (\Rightarrow_3) \to (\Rightarrow_4)$. Let $\sigma : U \Rightarrow_2 V$. Suppose further that $E[\uparrow]_1 \equiv E[\uparrow]_2$ for all E. Then*

$$E[\uparrow]_3[\sigma^\uparrow]_2 \equiv E[\sigma]_2[\uparrow]_1$$

for all E.

```
liftOp-up-mixed : ∀ {F} {G} {H} {F'} (comp₁ : Composition F G H) (comp₂ : Composition F' F H)
 {U} {V} {C} {K} {L} {σ : Op F U V} →
 (∀ {V} {C} {K} {L} {E : Subexp V C K} → ap F (up F {V} {L}) E ≡ ap F' (up F' {V} {L}) E) →
 ∀ {E : Subexp U C K} → ap F (liftOp F L σ) (ap G (up G) E) ≡ ap F' (up F') (ap F σ E)
```

Proof. Let $\circ_1 : (\Rightarrow_1); (\Rightarrow_2) \to (\Rightarrow_4)$ and $\circ_2 : (\Rightarrow_2); (\Rightarrow_3) \to (\Rightarrow_4)$. We have $E[\uparrow]_3[\sigma^\uparrow]_2 \equiv E[\sigma^\uparrow \circ_2 \uparrow]_4$ and $E[\sigma]_2[\uparrow]_1 \equiv E[\uparrow \circ_1 \sigma]_4$, so it is sufficient to prove that $\sigma^\uparrow \circ_2 \uparrow \sim \uparrow \circ_1 \sigma$.

We have

$$\begin{aligned}(\sigma^\uparrow \circ_2 \uparrow)(x) &\equiv \sigma^\uparrow(\uparrow(x)) \\ &\equiv \sigma(x)[\uparrow]_2 \\ &\equiv (\uparrow \circ_1 \sigma)(x)\square\end{aligned}$$

A.5 Family of Operations

Definition 18 (Family of Operations). *A* family of operations *consists of a pre-family with lifting $\Rightarrow$ and a composition $\circ : (\Rightarrow); (\Rightarrow) \to (\Rightarrow)$.*

```
record OpFamily : Set₂ where
 field
  liftFamily : LiftFamily
  comp : Composition liftFamily liftFamily liftFamily
 open LiftFamily liftFamily public
 open Composition comp public
```